
Resources

Course material

http://www.cs.uni-potsdam.de/wv/lehre

http://moodle.cs.uni-potsdam.de

http://potassco.sourceforge.net/teaching.html

Systems

clasp http://potassco.sourceforge.net

dlv http://www.dlvsystem.com

smodels http://www.tcs.hut.fi/Software/smodels

gringo http://potassco.sourceforge.net

lparse http://www.tcs.hut.fi/Software/smodels

clingo http://potassco.sourceforge.net

iclingo http://potassco.sourceforge.net

oclingo http://potassco.sourceforge.net

asparagus http://asparagus.cs.uni-potsdam.de

Literature

Books [4], [29], [53]

Surveys [50], [2], [39], [21], [11]

Articles [41], [42], [6], [61], [54], [49], [40], etc.

Modeling

For solving a problem class C for a problem instance I,
encode

1 the problem instance I as a set PI of facts and
2 the problem class C as a set PC of rules

such that the solutions to C for I can be (polynomially) extracted
from the stable models of PI ∪ PC

PI is (still) called problem instance

PC is often called the problem encoding

An encoding PC is uniform, if it can be used to solve all its
problem instances
That is, PC encodes the solutions to C for any set PI of facts

ASP solving process

Outline

ASP solving process

A case-study: Graph coloring

ASP solving process

Graph coloring

Problem instance A graph consisting of nodes and edges

facts formed by predicates node/1 and edge/2

Problem instance A graph consisting of nodes and edges

facts formed by predicates node/1 and edge/2

ASP solving process

Graph coloring

Problem instance A graph consisting of nodes and edges

facts formed by predicates node/1 and edge/2

facts formed by predicate col/1

Problem class Assign each node one color such that no two nodes
connected by an edge have the same color

ASP solving process

Graph coloring

Problem instance A graph consisting of nodes and edges

facts formed by predicates node/1 and edge/2

facts formed by predicate col/1

Problem class Assign each node one color such that no two nodes
connected by an edge have the same color

In other words,

1 Each node has a unique color
2 Two connected nodes must not have the same color

ASP solving process

Graph coloring

node(1..6).

edge(1,2). edge(1,3). edge(1,4).

edge(2,4). edge(2,5). edge(2,6).

edge(3,1). edge(3,4). edge(3,5).

edge(4,1). edge(4,2).

edge(5,3). edge(5,4). edge(5,6).

edge(6,2). edge(6,3). edge(6,5).

col(r). col(b). col(g).



Problem
instance

1 { color(X,C) : col(C) } 1 :- node(X).

:- edge(X,Y), color(X,C), color(Y,C).

 Problem
encoding

ASP solving process

ASP solving process

Graph coloring: Grounding

$ gringo --text color.lp

node(1). node(2). node(3). node(4). node(5). node(6).

edge(1,2). edge(1,3). edge(1,4). edge(2,4). edge(2,5). edge(2,6).

edge(3,1). edge(3,4). edge(3,5). edge(4,1). edge(4,2). edge(5,3).

edge(5,4). edge(5,6). edge(6,2). edge(6,3). edge(6,5).

col(r). col(b). col(g).

1 {color(1,r), color(1,b), color(1,g)} 1.

1 {color(2,r), color(2,b), color(2,g)} 1.

1 {color(3,r), color(3,b), color(3,g)} 1.

1 {color(4,r), color(4,b), color(4,g)} 1.

1 {color(5,r), color(5,b), color(5,g)} 1.

1 {color(6,r), color(6,b), color(6,g)} 1.

:- color(1,r), color(2,r). :- color(2,g), color(5,g). ... :- color(6,r), color(2,r).

:- color(1,b), color(2,b). :- color(2,r), color(6,r). :- color(6,b), color(2,b).

:- color(1,g), color(2,g). :- color(2,b), color(6,b). :- color(6,g), color(2,g).

:- color(1,r), color(3,r). :- color(2,g), color(6,g). :- color(6,r), color(3,r).

:- color(1,b), color(3,b). :- color(3,r), color(1,r). :- color(6,b), color(3,b).

:- color(1,g), color(3,g). :- color(3,b), color(1,b). :- color(6,g), color(3,g).

:- color(1,r), color(4,r). :- color(3,g), color(1,g). :- color(6,r), color(5,r).

:- color(1,b), color(4,b). :- color(3,r), color(4,r). :- color(6,b), color(5,b).

:- color(1,g), color(4,g). :- color(3,b), color(4,b). :- color(6,g), color(5,g).

:- color(2,r), color(4,r). :- color(3,g), color(4,g).

:- color(2,b), color(4,b). :- color(3,r), color(5,r).

:- color(2,g), color(4,g). :- color(3,b), color(5,b).

:- color(2,r), color(5,r). :- color(3,g), color(5,g).

:- color(2,b), color(5,b). :- color(4,r), color(1,r).

:- color(1,b), color(3,b). :- color(3,r), color(1,r). :- color(6,b), color(3,b).

ASP solving process

Graph coloring: Solving

$ gringo color.lp | clasp 0

clasp version 2.1.0

Reading from stdin

Solving...

Answer: 1

edge(1,2) ... col(r) ... node(1) ... color(6,b) color(5,g) color(4,b) color(3,r) color(2,r) color(1,g)

Answer: 2

edge(1,2) ... col(r) ... node(1) ... color(6,r) color(5,g) color(4,r) color(3,b) color(2,b) color(1,g)

Answer: 3

edge(1,2) ... col(r) ... node(1) ... color(6,g) color(5,b) color(4,g) color(3,r) color(2,r) color(1,b)

Answer: 4

edge(1,2) ... col(r) ... node(1) ... color(6,r) color(5,b) color(4,r) color(3,g) color(2,g) color(1,b)

Answer: 5

edge(1,2) ... col(r) ... node(1) ... color(6,g) color(5,r) color(4,g) color(3,b) color(2,b) color(1,r)

Answer: 6

edge(1,2) ... col(r) ... node(1) ... color(6,b) color(5,r) color(4,b) color(3,g) color(2,g) color(1,r)

SATISFIABLE

Models : 6

Time : 0.002s (Solving: 0.00s 1st Model: 0.00s Unsat: 0.00s)

CPU Time : 0.000s

ASP solving process

Problem

Logic
Program Grounder Solver Stable

Models

Solution

- - -

?

6

Modeling Interpreting

Solving

ASP solving process

A coloring

Answer: 6

edge(1,2) ... col(r) ... node(1) ... \

color(6,b) color(5,r) color(4,b) color(3,g) color(2,g) color(1,r)

1 2

3

4

5

6

ASP solving process

A coloring

Answer: 6

edge(1,2) ... col(r) ... node(1) ... \

color(6,b) color(5,r) color(4,b) color(3,g) color(2,g) color(1,r)

1 2

3

4

5

6

Methodology

Basic methodology

Methodology

Generate and Test (or: Guess and Check)

Generator Generate potential stable model candidates
(typically through non-deterministic constructs)

Tester Eliminate invalid candidates
(typically through integrity constraints)

Nutshell

Logic program = Data + Generator + Tester ( + Optimizer)

Methodology Satisfiability

Satisfiability testing

Problem Instance: A propositional formula φ in CNF

Problem Class: Is there an assignment of propositional variables to
true and false such that a given formula φ is true

Example: Consider formula

(a ∨ ¬b) ∧ (¬a ∨ b)

Logic Program:

Generator Tester Stable models
{ a, b } ← ← ∼a, b

← a,∼b
X1 = {a, b}
X2 = {}

Methodology Queens

The n-Queens Problem

5 Z0Z0Z
4 0Z0Z0
3 Z0Z0Z
2 0Z0Z0
1 Z0Z0Z

1 2 3 4 5

Place n queens on an n × n
chess board

Queens must not attack one
another

Q Q Q

Q Q

M. Gebser and T. Schaub (KRR@UP) Answer Set Solving in Practice July 13, 2013 26 / 52



Methodology Queens

Defining the Field

queens.lp

row(1..n).

col(1..n).

Create file queens.lp

Define the field

n rows
n columns

Methodology Queens

Defining the Field

Running . . .

$ gringo queens.lp --const n=5 | clasp

Answer: 1

row(1) row(2) row(3) row(4) row(5) \

col(1) col(2) col(3) col(4) col(5)

SATISFIABLE

Models : 1

Time : 0.000

Prepare : 0.000

Prepro. : 0.000

Solving : 0.000

Methodology Queens

Placing some Queens

queens.lp

row(1..n).

col(1..n).

{ queen(I,J) : row(I) : col(J) }.

Guess a solution candidate

by placing some queens on the board

Methodology Queens

Placing some Queens

Running . . .

$ gringo queens.lp --const n=5 | clasp 3

Answer: 1

row(1) row(2) row(3) row(4) row(5) \

col(1) col(2) col(3) col(4) col(5)

Answer: 2

row(1) row(2) row(3) row(4) row(5) \

col(1) col(2) col(3) col(4) col(5) queen(1,1)

Answer: 3

row(1) row(2) row(3) row(4) row(5) \

col(1) col(2) col(3) col(4) col(5) queen(2,1)

SATISFIABLE

Models : 3+

...

Methodology Queens

Placing some Queens: Answer 1

Answer 1

5 Z0Z0Z
4 0Z0Z0
3 Z0Z0Z
2 0Z0Z0
1 Z0Z0Z

1 2 3 4 5

Methodology Queens

Placing some Queens: Answer 2

Answer 2

5 Z0Z0Z
4 0Z0Z0
3 Z0Z0Z
2 0Z0Z0
1 L0Z0Z

1 2 3 4 5

Methodology Queens

Placing some Queens: Answer 3

Answer 3

5 Z0Z0Z
4 0Z0Z0
3 Z0Z0Z
2 QZ0Z0
1 Z0Z0Z

1 2 3 4 5

Methodology Queens

Placing n Queens

queens.lp

row(1..n).

col(1..n).

{ queen(I,J) : row(I) : col(J) }.

:- not n { queen(I,J) } n.

Place exactly n queens on the board

Methodology Queens

Placing n Queens

Running . . .

$ gringo queens.lp --const n=5 | clasp 2

Answer: 1

row(1) row(2) row(3) row(4) row(5) \

col(1) col(2) col(3) col(4) col(5) \

queen(5,1) queen(4,1) queen(3,1) \

queen(2,1) queen(1,1)

Answer: 2

row(1) row(2) row(3) row(4) row(5) \

col(1) col(2) col(3) col(4) col(5) \

queen(1,2) queen(4,1) queen(3,1) \

queen(2,1) queen(1,1)

...

Methodology Queens

Placing n Queens: Answer 1

Answer 1

5 L0Z0Z
4 QZ0Z0
3 L0Z0Z
2 QZ0Z0
1 L0Z0Z

1 2 3 4 5

Methodology Queens

Placing n Queens: Answer 2

Answer 2

5 Z0Z0Z
4 QZ0Z0
3 L0Z0Z
2 QZ0Z0
1 LQZ0Z

1 2 3 4 5

Methodology Queens

Horizontal and Vertical Attack

queens.lp

row(1..n).

col(1..n).

{ queen(I,J) : row(I) : col(J) }.

:- not n { queen(I,J) } n.

:- queen(I,J), queen(I,JJ), J != JJ.

:- queen(I,J), queen(II,J), I != II.

Forbid horizontal attacks

Forbid vertical attacks

Methodology Queens

Horizontal and Vertical Attack

Running . . .

$ gringo queens.lp --const n=5 | clasp

Answer: 1

row(1) row(2) row(3) row(4) row(5) \

col(1) col(2) col(3) col(4) col(5) \

queen(5,5) queen(4,4) queen(3,3) \

queen(2,2) queen(1,1)

...

Methodology Queens

Horizontal and Vertical Attack: Answer 1

Answer 1

5 Z0Z0L
4 0Z0L0
3 Z0L0Z
2 0L0Z0
1 L0Z0Z

1 2 3 4 5

Methodology Queens

Diagonal Attack

queens.lp

row(1..n).

col(1..n).

{ queen(I,J) : row(I) : col(J) }.

:- not n { queen(I,J) } n.

:- queen(I,J), queen(I,JJ), J != JJ.

:- queen(I,J), queen(II,J), I != II.

:- queen(I,J), queen(II,JJ), (I,J) != (II,JJ), I-J == II-JJ.

:- queen(I,J), queen(II,JJ), (I,J) != (II,JJ), I+J == II+JJ.

Forbid diagonal attacks

Methodology Queens

Diagonal Attack

Running . . .

$ gringo queens.lp --const n=5 | clasp

Answer: 1

row(1) row(2) row(3) row(4) row(5) \

col(1) col(2) col(3) col(4) col(5) \

queen(4,5) queen(1,4) queen(3,3) queen(5,2) queen(2,1)

SATISFIABLE

Models : 1+

Time : 0.000

Prepare : 0.000

Prepro. : 0.000

Solving : 0.000

Methodology Queens

Diagonal Attack: Answer 1

Answer 1

5 ZQZ0Z
4 0Z0ZQ
3 Z0L0Z
2 QZ0Z0
1 Z0ZQZ

1 2 3 4 5

Methodology Queens

Optimizing

queens-opt.lp

1 { queen(I,1..n) } 1 :- I = 1..n.

1 { queen(1..n,J) } 1 :- J = 1..n.

:- 2 { queen(D-J,J) }, D = 2..2*n.

:- 2 { queen(D+J,J) }, D = 1-n..n-1.

Encoding can be optimized

Much faster to solve

Methodology Queens

And sometimes it rocks
$ clingo -c n=5000 queens-opt-diag.lp --config=jumpy -q --stats=3

clingo version 4.1.0

Solving...

SATISFIABLE

Models : 1+

Time : 3758.143s (Solving: 1905.22s 1st Model: 1896.20s Unsat: 0.00s)

CPU Time : 3758.320s

Choices : 288594554

Conflicts : 3442 (Analyzed: 3442)

Restarts : 17 (Average: 202.47 Last: 3442)

Model-Level : 7594728.0

Problems : 1 (Average Length: 0.00 Splits: 0)

Lemmas : 3442 (Deleted: 0)

Binary : 0 (Ratio: 0.00%)

Ternary : 0 (Ratio: 0.00%)

Conflict : 3442 (Average Length: 229056.5 Ratio: 100.00%)

Loop : 0 (Average Length: 0.0 Ratio: 0.00%)

Other : 0 (Average Length: 0.0 Ratio: 0.00%)

Atoms : 75084857 (Original: 75069989 Auxiliary: 14868)

Rules : 100129956 (1: 50059992/100090100 2: 39990/29856 3: 10000/10000)

Bodies : 25090103

Equivalences : 125029999 (Atom=Atom: 50009999 Body=Body: 0 Other: 75020000)

Tight : Yes

Variables : 25024868 (Eliminated: 11781 Frozen: 25000000)

Constraints : 66664 (Binary: 35.6% Ternary: 0.0% Other: 64.4%)

Backjumps : 3442 (Average: 681.19 Max: 169512 Sum: 2344658)

Executed : 3442 (Average: 681.19 Max: 169512 Sum: 2344658 Ratio: 100.00%)

Bounded : 0 (Average: 0.00 Max: 0 Sum: 0 Ratio: 0.00%)

Methodology Traveling Salesperson

Traveling Salesperson

node(1..6).

edge(1,2;3;4). edge(2,4;5;6). edge(3,1;4;5).

edge(4,1;2). edge(5,3;4;6). edge(6,2;3;5).

cost(1,2,2). cost(1,3,3). cost(1,4,1).

cost(2,4,2). cost(2,5,2). cost(2,6,4).

cost(3,1,3). cost(3,4,2). cost(3,5,2).

cost(4,1,1). cost(4,2,2).

cost(5,3,2). cost(5,4,2). cost(5,6,1).

cost(6,2,4). cost(6,3,3). cost(6,5,1).

Methodology Traveling Salesperson

Traveling Salesperson

node(1..6).

edge(1,2;3;4). edge(2,4;5;6). edge(3,1;4;5).

edge(4,1;2). edge(5,3;4;6). edge(6,2;3;5).

cost(1,2,2). cost(1,3,3). cost(1,4,1).

cost(2,4,2). cost(2,5,2). cost(2,6,4).

cost(3,1,3). cost(3,4,2). cost(3,5,2).

cost(4,1,1). cost(4,2,2).

cost(5,3,2). cost(5,4,2). cost(5,6,1).

cost(6,2,4). cost(6,3,3). cost(6,5,1).

M. Gebser and T. Schaub (KRR@UP) Answer Set Solving in Practice July 13, 2013 47 / 52



Methodology Traveling Salesperson

Traveling Salesperson

1 { cycle(X,Y) : edge(X,Y) } 1 :- node(X).

1 { cycle(X,Y) : edge(X,Y) } 1 :- node(Y).

reached(Y) :- cycle(1,Y).

reached(Y) :- cycle(X,Y), reached(X).

:- node(Y), not reached(Y).

#minimize [ cycle(X,Y) = C : cost(X,Y,C) ].

Methodology Traveling Salesperson

Methodology Reviewer Assignment

Reviewer Assignment
by Ilkka Niemelä

reviewer(r1). paper(p1). classA(r1,p1). classB(r1,p2). coi(r1,p3).

reviewer(r2). paper(p2). classA(r1,p3). classB(r1,p4). coi(r1,p6).

...

3 { assigned(P,R) : reviewer(R) } 3 :- paper(P).

:- assigned(P,R), coi(R,P).

:- assigned(P,R), not classA(R,P), not classB(R,P).

:- not 6 { assigned(P,R) : paper(P) } 9, reviewer(R).

assignedB(P,R) :- classB(R,P), assigned(P,R).

:- 3 { assignedB(P,R) : paper(P) }, reviewer(R).

#minimize { assignedB(P,R) : paper(P) : reviewer(R) }.

Methodology Planning

Simplistic STRIPS Planning

time(1..k). lasttime(T) :- time(T), not time(T+1).

fluent(p). action(a). action(b). init(p).

fluent(q). pre(a,p). pre(b,q).

fluent(r). add(a,q). add(b,r). query(r).

del(a,p). del(b,q).

holds(P,0) :- init(P).

1 { occ(A,T) : action(A) } 1 :- time(T).

:- occ(A,T), pre(A,F), not holds(F,T-1).

holds(F,T) :- holds(F,T-1), not nolds(F,T), time(T).

holds(F,T) :- occ(A,T), add(A,F).

nolds(F,T) :- occ(A,T), del(A,F).

:- query(F), not holds(F,T), lasttime(T).

In V. Dahl and I. Niemelä, editors, Proceedings of the Twenty-third
International Conference on Logic Programming (ICLP’07), volume
4670 of Lecture Notes in Computer Science, pages 119–133.
Springer-Verlag, 2007.

[39] M. Gelfond.
Answer sets.
In V. Lifschitz, F. van Harmelen, and B. Porter, editors, Handbook of
Knowledge Representation, chapter 7, pages 285–316. Elsevier
Science, 2008.

[40] M. Gelfond and N. Leone.

M. Gebser and T. Schaub (KRR@UP) Answer Set Solving in Practice July 13, 2013 52 / 52



Logic programming and knowledge representation — the A-Prolog
perspective.
Artificial Intelligence, 138(1-2):3–38, 2002.

[41] M. Gelfond and V. Lifschitz.
The stable model semantics for logic programming.
In R. Kowalski and K. Bowen, editors, Proceedings of the Fifth
International Conference and Symposium of Logic Programming
(ICLP’88), pages 1070–1080. MIT Press, 1988.

[42] M. Gelfond and V. Lifschitz.
Logic programs with classical negation.
In D. Warren and P. Szeredi, editors, Proceedings of the Seventh
International Conference on Logic Programming (ICLP’90), pages
579–597. MIT Press, 1990.

[43] E. Giunchiglia, Y. Lierler, and M. Maratea.
Answer set programming based on propositional satisfiability.
Journal of Automated Reasoning, 36(4):345–377, 2006.

M. Gebser and T. Schaub (KRR@UP) Answer Set Solving in Practice July 13, 2013 52 / 52



[44] P. Hill and D. Warren, editors.
Proceedings of the Twenty-fifth International Conference on Logic
Programming (ICLP’09), volume 5649 of Lecture Notes in Computer
Science. Springer-Verlag, 2009.

[45] J. Huang.
The effect of restarts on the efficiency of clause learning.
In Veloso [68], pages 2318–2323.

[46] K. Konczak, T. Linke, and T. Schaub.
Graphs and colorings for answer set programming.
Theory and Practice of Logic Programming, 6(1-2):61–106, 2006.

[47] J. Lee.
A model-theoretic counterpart of loop formulas.
In L. Kaelbling and A. Saffiotti, editors, Proceedings of the
Nineteenth International Joint Conference on Artificial Intelligence
(IJCAI’05), pages 503–508. Professional Book Center, 2005.

M. Gebser and T. Schaub (KRR@UP) Answer Set Solving in Practice July 13, 2013 52 / 52



[48] N. Leone, G. Pfeifer, W. Faber, T. Eiter, G. Gottlob, S. Perri, and
F. Scarcello.
The DLV system for knowledge representation and reasoning.
ACM Transactions on Computational Logic, 7(3):499–562, 2006.

[49] V. Lifschitz.
Answer set programming and plan generation.
Artificial Intelligence, 138(1-2):39–54, 2002.

[50] V. Lifschitz.
Introduction to answer set programming.
Unpublished draft, 2004.

[51] V. Lifschitz and A. Razborov.
Why are there so many loop formulas?
ACM Transactions on Computational Logic, 7(2):261–268, 2006.

[52] F. Lin and Y. Zhao.
ASSAT: computing answer sets of a logic program by SAT solvers.
Artificial Intelligence, 157(1-2):115–137, 2004.

M. Gebser and T. Schaub (KRR@UP) Answer Set Solving in Practice July 13, 2013 52 / 52



[53] V. Marek and M. Truszczyński.

